Basic math - graphs and polynomial

Robert Mařík

27. listopadu 2010

1 Plotting functions

We can plot a function, or add two or more plots using `plot` and `show` commands

Sage code
```
plot(x^2,(x,-1,1))
```



Sage code
```
P1=plot(x^2,(x,-3,3))
P2=plot(tan(x),(x,-6,6),color='red')
show(P1+P2,ymin=-6,ymax=6)
```

[0]Podporováno grantem FRVŠ 131/2010.



On the previous picture we can see that Sage draws vertical lines at the points of discontinuity. This can be avoided by using `detect_poles` directive.

Sage code
```
P2=plot(tan(x),(x,-6,6),color='green',detect_poles=True)
show(P1+P2,ymin=-pi,ymax=pi)
```



2 Polynomials

Sage is capable to simplify expressions. Among others, we can simplify polynomials (factor, expand), divide polynomials and look for zeros of polynomials. The underscore in the following computations reffers allways to previous result.

Sage code
```
P(x)=(x-2)^3*(x^2-4*x+3)*2
P(x)
```

$2\,(x-2)^3(x^2-4\,x+3)$

Here we multiply parentheses and factor again.

Sage code

```
expand(_)
```

$2\,x^5 - 20\,x^4 + 78\,x^3 - 148\,x^2 + 136\,x - 48$

Sage code

```
factor(_)
```

$2\,(x-3)(x-2)^3(x-1)$

We can substitute values for x.

Sage code

```
[P(0),P(1),P(2),P(sqrt(2))]
```

$\left[-48, 0, 0, -2\,(\sqrt{2}-2)^3(4\,\sqrt{2}-5)\right]$

The last value can be computed numerically.

Sage code

```
P(sqrt(2)).n()
```

0.264068711928515

Here we divide and simplify the quotient. If we divide by the factor $(x-c)$, where c is a zero of the polynomial, the remainder is zero.

Sage code

```
(P(x)/(x-1))
```

$2\,\frac{(x-2)^3(x^2-4\,x+3)}{(x-1)}$

Sage code

```
(P(x)/(x-1)).simplify_full()
```

$2\,x^4 - 18\,x^3 + 60\,x^2 - 88\,x + 48$

Sage code

```
P(x)._maxima_().divide(x-1).sage()
```

$\left[2\,x^4 - 18\,x^3 + 60\,x^2 - 88\,x + 48, 0\right]$

Sage code

```
P(x)._maxima_().divide(x-2).sage()
```

$\left[2\,x^4 - 16\,x^3 + 46\,x^2 - 56\,x + 24, 0\right]$

If c is not a zero of the polynomial, then we get nonzero remainder. This remainder is equal to the value of the function $P(x)$ at c. We test this property for $c = 5$.

Sage code

```
P(x)._maxima_().divide(x-5).sage()
```

$\left[2\,x^4 - 10\,x^3 + 28\,x^2 - 8\,x + 96, 432\right]$

Sage code

```
P(5)
```

432

Another method how to divide polynomials in Sage is to define the ring of polynomials, convert symbolic expression into polynomial and use `quo_rem` method for dividing polynomials.
Here W is ring of polynomials with variable x and $W(P(x))$ means "treat $P(x)$ as polynomial".

Sage code

```
W.<x>=QQ[]
W(P(x)).quo_rem(x-5)
```

$\left(2x^4 - 10x^3 + 28x^2 - 8x + 96, 432\right)$

3 Multiplicity

We define polynomial

Sage code
```
P(x)=x^5 + 9*x^4 + 36*x^3 + 80*x^2 + 96*x + 48
P(x)
```

$x^5 + 9\,x^4 + 36\,x^3 + 80\,x^2 + 96\,x + 48$

$x = -2$ is a zero of this polynomial.

Sage code
```
P(-2)
```

0

Thus by Theorem of Bezout, we can divide the polynomial $P(x)$ by the polynomial $x + 2$.

Sage code
```
(P(x)/(x+2)).simplify_full()
```

$x^4 + 7\,x^3 + 22\,x^2 + 36\,x + 24$

Let us try to perform the division once more.

Sage code
```
(_/(x+2)).simplify_full()
```

$x^3 + 5\,x^2 + 12\,x + 12$

.. and once more ....

Sage code
```
(_/(x+2)).simplify_full()
```

$x^2 + 3\,x + 6$

... and once more ...

Sage code
```
(_/(x+2)).simplify_full()
```

$\frac{(x^2 + 3\,x + 6)}{(x + 2)}$

The last division cannot be performed without remainder. Really, the polynomial $x^2 + 3x + 6$ has not zero $x = -2$. To summarize, we succeded to divide the polynomial $P(x)$ by $x + 2$ three times. The last quotient is $x^2 + 3x + 6$. Hence $P(x)$ can be written as $(x + 2)^3(x^2 + 3x + 6)$.

Sage code
```
((x+2)^3*(x^2+3*x+6)).show()
((x+2)^3*(x^2+3*x+6)).expand()
```

$$(x + 2)^3 (x^2 + 3\,x + 6)$$

$x^5 + 9\,x^4 + 36\,x^3 + 80\,x^2 + 96\,x + 48$

We can ensure that the polynomial equals $P(x)$. The underscore reffers to the previous result and `bool` function tests, if the equality is True.

Sage code
```
bool((_) == P(x))
```

True

4 Multiplicity and sign of the polynomial

Sage code
```
P(x)=(x-2)^2*x^3*(x+1)
P(x)
```

$(x-2)^2(x+1)x^3$

The polynomial changes sing in the neigborhood of the zero with odd multiplicity and preserves sign in the neighborhood of the zero with even multiplicity.

Sage code
```
plot(P(x),(x,-2,3)).show(ymin=-1,ymax=4,figsize=4)
```